MATH 8020 Minimum Weight Matchings examples in SAS

The linear assignment function finds a matching of minimum weight in a bipartite graph. In this first example, we create a graph of edge weights in matrix form. This bipartite graph creates an edge between swimmers and their average time with a particular swimming stroke. Since faster times are better in competitive swimming, we want a matching of minimum total weight of swimmers to strokes.

```
data RelayTimesMatrix;
input name $ sex $ back breast fly free;
* matrix form of a graph;
datalines;
Sue     F 40.9 36.7 28.3 36.1
Karen   F 34.6 33.0 28.2 29.6
Jan     F 31.3 41.9 27.1 26.2
Andrea  F 28.6 43.9 29.1 27.0
Pat     F 18.9 35.1 26.6 25.2
Carol   F 32.9 44.7 26.6 29.3
;
proc optgraph
data_matrix = RelayTimesMatrix;
linear_assignment
out      = swimmers_to_stroke
weight   = (back--free)
id       = (name sex);
run;
```

The optimal matching is sent to the file swimmers_to_stroke and the sum of the weights appears in the results file.

It seems natural to want to find a maximum weight matching in certain cases. Historically speaking, maximum weight matchings are introduced as a marriage problem. Different pairs of women and men have different compatibility ratings. The goal is to create a matching that maximizes total compatibility. Matching employees to tasks at work is a similar application that calls for a maximum weight matching. To employ the minimum weight linear assignment call, merely subtract each compatibility rating from a fixed constant to create and incompatibility rating. Minimizing incompatibility will automatically maximize compatibility.

In the next example, we match members of the Beatles and the Go-Go's for minimum incompatibility.

```
data MarriageLinks;
input woman $ man $ incompatibility ;
* edge weight form of graph;
datalines;
Belinda John   36.7
Belinda George 28.3
Belinda Paul   36.1
Kathy George 34.6
Kathy John   26.2
```

```
Jane George 31.3
Jane Paul   27.1
Gina George 28.6
Gina Paul  29.1
Charlotte Ringo  32.9
Charlotte Paul 26.6
;
proc optgraph
graph_direction = directed
data_links      = MarriageLinks;
data_links_var
from         = woman
to           = man
weight       = incompatibility ;
linear_assignment
out          = marriage_pref;
run;
```

As before, the optimal matching is sent to the file marriage_pref and the sum of the weights appears in the results file.